Beam with Prescribed End Displacements

Here we consider a beam without any distributed load, as shown in the figure below. Instead the beam has prescribed displacement and rotation at both ends. It is emphasized that these are SMALL deformations, so we can use the basic Euler-Bernoulli beam theory. The objective is to plot the displaced shape, as well as the bending moment and shear force diagrams.



Input values in N and mm

The cross-section is rectangular and composed of three "two by ten" dimension lumber members placed side-by-side. The total width is denoted *b* and the height *h*.

```
indepInput = {b → (3 × 38), h → 235, L → 4000, E → 9500};
depInput = {I → b h^3 / 12};
```

The imposed deformations are:

```
wA = 10;
θA = 2 / 180 * π;
wB = -20;
θB = 1 / 180 * π;
```

Solving the differential equation

Because the beam does not have any external load the differential equation is $\frac{d^4 w}{dx^4} = 0$. The general solution is obtained by integrating that differential equation four times, which yields:

```
w = C1 x^3 + C2 x^2 + C3 x + C4;
```

The boundary conditions given by the imposed deformations yield four equations in the four unknowns:

```
Eq1 = (w /. x → 0) == wA;
Eq2 = (w /. x → L) == wB;
Eq3 = (∂x w /. x → 0) == -θA;
Eq4 = (∂x w /. x → L) == -θB;
```

The joint solution of those four equations gives the value of the integration constants:

```
sol = Solve[{Eq1, Eq2, Eq3, Eq4}, {C1, C2, C3, C4}]
```

which yields: $\left\{\left\{C1 \to -\frac{-3600 + L\pi}{60\,L^3},\ C2 \to -\frac{3240 - L\pi}{36\,L^2},\ C3 \to -\frac{\pi}{90},\ C4 \to 10\right\}\right\}$

Substitute the solution, as well as the input values, into the expressions for *w*, *θ*, *M*, and *V*:

```
w = w /. sol /. depInput /. indepInput;
θ = ∂x w /. sol /. depInput /. indepInput;
M = E I ∂x ∂x w /. sol /. depInput /. indepInput;
V = E I ∂x ∂x ∂x w /. sol /. depInput /. indepInput;
```

Plots of that solution:

```
Plot[w, {x, 0, L /. indepInput}, PlotLabel → "Displacement, w",
 PlotRange → Full, PlotStyle → Black]
```



```
Plot[θ, {x, 0, L /. indepInput}, PlotLabel → "Rotation, θ",
 PlotRange → Full, PlotStyle → Black]
```




```
Plot[M, {x, 0, L /. indepInput}, PlotLabel → "Moment, M",
 PlotRange → Full, PlotStyle → Black]
```



```
Plot[V, {x, 0, L /. indepInput}, PlotLabel → "Shear force, V",
 PlotRange → Full, PlotStyle → Black]
```